Newsgroup: it.comp.appl.access
Titolo: aiutooo... urgentisss... cavolo è già tardi... MDE e tasto destro

Ho letto di vedere sul sitocomune per fare il file mde con annessa la
protezione dalla pressione di shift all'avvio ma non trovo nulla.... o
faccio l'MDE o lo proteggo... come faccio a combinare le due cose?
Sono agli sgoccioli...
Ciao
Alessio

 ha scritto nel messaggio
news:b485s4$oge$1...@newsfeed.cineca.it...

Lo Shift non protegge nulla, basta una Routinetta stupidissima
per eliminarlo....!
Se vuoi proteggerlo fai la versione MDE oppure protezione Utente.
@Alex.

 ha scritto nel messaggio
news:0bN9a.183041$YG2.5...@twister1.libero.it...

Si lo so ma anche in MDE riescono ad accedere alle tabelle e esportarle...
Io volevo un banale metodo per non far esportare il contenuto delle
tabelle...

 ha scritto nel messaggio
news:b485s4$oge$1...@newsfeed.cineca.it...

Personalmente utilizzo queste 2 procedure all'interno di un form di "Accesso
Alla Struttura", nel tuo caso la proprietą che stai cercando č
"AllowBypassKey"
Per progetti ADP
******************************************
Private Sub SetStartupProperties(Lock As Boolean)
    Dim Prps As AccessObjectProperties
    Const AppCap As String = "AppCap"
    Const FrmLock As String = "FrmLock"
    Const FrmUnLock As String = "FrmUnLock"
    Const FileLogo As String = "FileLogo.ico"
    Const BarraMenu As String = "BarraMenu"
    Const VisualizzaBarraDiStato As Boolean = True
    On Error GoTo ERRORE
    Set Prps = CurrentProject.Properties
    Call SetProperty(Prps, "StartupForm", IIf(Lock, FrmUnLock, FrmLock))
    Call SetProperty(Prps, "StartUpMenuBar", IIf(Lock, "(Predefinita)",
BarraMenu))
    Call SetProperty(Prps, "AppIcon", CurrentProject.Path & "\" & FileLogo)
    Call SetProperty(Prps, "AppTitle", AppCap)
    Call SetProperty(Prps, "StartUpShowDBWindow", Lock)
    'Call SetProperty(Prps, "StartUpShowStatusBar", IIf(Lock, Lock,
VisualizzaBarraDiStato))
    Call SetProperty(Prps, "AllowShortcutMenus", Lock)
    Call SetProperty(Prps, "AllowFullMenus", Lock)
    Call SetProperty(Prps, "AllowBuiltInToolbars", Lock)
    Call SetProperty(Prps, "AllowToolbarChanges", Lock)
    Call SetProperty(Prps, "AllowBypassKey", Lock)
    Call SetProperty(Prps, "AllowSpecialKeys", Lock)
    Set Prps = Nothing
    Application.Quit
ESCI:
    Exit Sub
ERRORE:
    Call
omwHandleError("AccessoAllaStruttura.SetStartupProperties",Err.Description,E
rr.Number)
    Resume ESCI
End Sub
Private Sub SetProperty(Prps As AccessObjectProperties, Nome As String,
Valore)
    On Error GoTo ERRORE
    Prps.Item(Nome).Value = Valore
ESCI:
    Exit Sub
ERRORE:
    If Err.Number = 2455 Then
        Prps.Add Nome, Valore
        Resume Next
    Else
        Call
omwHandleError("AccessoAllaStruttura.SetProperty",Err.Description,Err.Number
)
    End If
    Resume ESCI
End Sub
******************************************
Mentre per database MDB
******************************************
Private Sub SetStartupProperties(Lock As Boolean)
    Const DB_Text As Long = 10
    Const DB_Boolean As Long = 1
    Const AppCap As String = "AppCap"
    Const FrmLock As String = "FrmLock"
    Const FrmUnLock As String = "FrmUnLock"
    Const FileLogo As String = "FileLogo.ico"
    Const BarraMenu As String = "BarraMenu"
    Const VisualizzaBarraDiStato As Boolean = True
    On Error GoTo ERRORE
    ChangeProperty "StartupForm", DB_Text, IIf(Lock, FrmUnLock, FrmLock)
    ChangeProperty "StartUpMenuBar", DB_Text, IIf(Lock, "(Predefinita)",
BarraMenu)
    ChangeProperty "AppIcon", DB_Text, CurrentProject.Path & "\" & FileLogo
    ChangeProperty "AppTitle", DB_Text, AppCap
    ChangeProperty "StartUpShowDBWindow", DB_Boolean, Lock
    ChangeProperty "StartUpShowStatusBar", DB_Boolean, IIf(Lock, Lock,
VisualizzaBarraDiStato))
    ChangeProperty "AllowShortcutMenus", DB_Boolean, Lock
    ChangeProperty "AllowFullMenus", DB_Boolean, Lock
    ChangeProperty "AllowBuiltInToolbars", DB_Boolean, Lock
    ChangeProperty "AllowToolbarChanges", DB_Boolean, Lock
    ChangeProperty "AllowBypassKey", DB_Boolean, Lock
    ChangeProperty "AllowSpecialKeys", DB_Boolean, Lock
ESCI:
    Exit Sub
ERRORE:
    Call
omwHandleError("AccessoAllaStruttura.SetProperty",Err.Description,Err.Number
)
    Resume ESCI
End Sub
Private Sub ChangeProperty(PropName As String, PropType As Variant,
PropValue As Variant)
    Dim dbs As Object, prp As Variant
    Const conPropNotFoundError = 3270
    On Error GoTo ERRORE
    Set dbs = CurrentDb
    dbs.Properties(PropName) = PropValue
ESCI:
    Exit Sub
ERRORE:
    If Err = conPropNotFoundError Then
        Set prp = dbs.CreateProperty(PropName, PropType, PropValue)
        dbs.Properties.Append prp
        Resume Next
    Else
        Call
omwHandleError("AccessoAllaStruttura.SetProperty",Err.Description,Err.Number
)
        Resume ESCI
    End If
End Sub
******************************************
Ciao Omnitow

Grrrazie a tutti!!
Ale



prova a nasconderle e a disabilitare l'uso dei menu per rifarle vedere
Saluti.
Emanuele.